

(a) A directed graph

(0,1) (2,4) (6,1)
(1,2) (3,4) (6,7)
(1,3) (3,5) (4,6)
(2,2) (4,0) (5,6)

(b) Edge list

To Node

From Node	0	1	2	3	4	5	6	7
0	0	1	0	0	0	0	0	0
1	0	0	1	1	0	0	0	0
2	0	0	1	0	1	0	0	0
3	0	0	0	0	1	1	0	0
4	1	0	0	0	0	0	1	0
5	0	0	0	0	0	0	1	0
6	0	1	0	0	0	0	0	1
7	0	0	0	0	0	0	0	0

(c) Adjacency matrix

Figure 1: An example of a directed graph and the corresponding edge list and adjacency matrix.

Question 4 (30 points):

Representation of a Directed Graph: A directed graph $G(V, E)$ is formed by a set of vertices V and a set of edges E. A vertex is often also called a "node" in the graph. Two common data representations for a directed graph are a list of edges and an adjacency matrix. In this question, we assume that the vertices are labeled with integer values: 0, 1, 2, $\cdots$. Figure 1 shows an example of a directed graph with the corresponding edge list and adjacency matrix to illustrate these two data representation. The graph shown in Figure 1 is only an example. This question asks that you write code that works for **ANY** directed graph.

Representation of an edge list in memory: In this question, the edge list is represented as a list of 32-bit words. Two words are used to represent each edge. The first word contains the number of the node that is the source of the edge, also called the `fromNode`, the second word contains the number of the node that is the target of the edge, also called the `toNode`. The end of the list is signalled by a sentinel value, which is a 32-bit word containing the value -1 (`0xFFFF FFFF`).

Representation of the Adjacency Matrix in memory: Consider a directed graph that has k nodes. The adjacency matrix will be represented by k bit vectors, each vector containing k bits. The bit vectors will be stored sequentially in memory. Let w be the number of words occupied by each bit vector. k is determined by the number of nodes in the graph. If $k \leq 32$, then $w = 1$. If $33 \leq k \leq 64$ then $w = 2$, *etc.* In general:

$$w = \left\lceil \frac{k}{32} \right\rceil \tag{1}$$

In this question you will write MIPS assembly to convert an edge-list representation of a directed graph into an adjacency-matrix representation of the same graph. Your code will be divided into two functions as specified below. You must write the two functions to work independently and you must follow all the MIPS calling conventions in both of them.

1. (**15 points**) Write the MIPS assembly code for a function called `SetBit` that receives the address in memory where a bit vector is stored and the position of a bit to be set to 1. This function must work for any bit-vector length (the bit position may be higher than 32). Therefore a bit vector may be formed by multiple words. Within each of the words of a bit vector, the lowest bit position will correspond to the Least-Significant Bit of the word (bit 0). The only change that this function does is to write the bit 1 is the specified bit position.

   Input:

   `$a0`: address of first word of bit vector `$a1`: bit to be set

```
34 # SetBit:
35 # Parameters:
36 #   $a0: address of vector to be changed
37 #   $a1: bit to be set
38 SetBit:
39     srl     $t0, $a1, 5         # $t0 <- floor(bit/32)
40     sll     $t1, $t0, 2         # $t1 <- word offset
41     add     $t2, $a0, $t1       # $t2 <- word address
42     lw      $t3, 0($t2)         # $t3 <- word
43     andi    $t4, $a1, 0x001F    # $t4 <- bit%32
44     li      $t5, 1              # $t5 <- 1
45     sllv    $t6, $t5, $t4       # $t6 <- mask with correct bit 1
46     or      $t7, $t3, $t6       # $t7 <- word with bit set
47     sw      $t7, 0($t2)         # Write word back to memory
48     jr      $ra
```

2. (**15 points**) Write MIPS assembly code for a function called `ConvertGraph` that converts the representation of a directed graph from a list of edges to an adjacency-matrix representation. The adjacency matrix is formed by k bit vectors that are stored contiguously in memory, where k is the number of nodes in the graph. The interface for this routine is as follows:

   Input: A directed graph specified as an edge list stored in memory.

   Output: An adjacenty-matrix representation of the same directed graph, stored in memory.

   Library: A collection of bit-vector routines that include a `SetBit` routine.

   Precondition: The adjacency matrix is pre-allocated and zeroed.

   Parameters: These are the input parameters for the function:

   - `$a0`: address of first word of first edge in edge list
   - `$a1`: number of nodes in directed graph
   - `$a2`: address of first bit vector in adjacency matrix

```
# ConvertGraph
# Input: A directed graph specified as an edge list
# Output: The same directed graph specified as an adjacency matrix
    represented by a set of binary vectors
# Library:
#   A collection of bit-vector routines that include a SetBit routine
# Precondition:
#   Adjacency matrix is pre-allocated and zeroed.
# Parameters:
#    $a0: address of first word of first edge in edge list
#    $a1: number of nodes in graph
#    $a2: address of first bit vector in adjacency matrix
.text
ConvertGraph:
    add     $sp, $sp, -20
    sw      $ra, 0($sp)
    sw      $s0, 4($sp)
    sw      $s1, 8($sp)
    sw      $s2, 12($sp)
    sw      $s3, 16($sp)
    li      $s0, -1             # $s0 <- -1
    move    $s1, $a0            # $s1 <- Address of first edge
    move    $s2, $a2            # base address for bit vectors
    srl     $s3, $a1, 5         # $s3 <- floor(numberNodes/32)
    andi    $t0, $a1, 0x001F    # $t0 <- numberNodes%32
    beq     $t0, $0, NextEdge
    addi    $s3, $s3, 1         # $s3 <- #WordsPerBitVector
NextEdge:
    lw      $t1, 0($s1)         # $t1 <- fromNode
    beq     $t1, $s0, Done      # if fromNode = -1 goto Done
    mul     $t2, $t1, $s3       # $t2 <- fromNode x #WordsPerBitVector
    sll     $t3, $t2, 2         # $t3 <- Offset into AdjacencyMatrix
# Call setBit(BitVectorAddress,ToNode)
    add     $a0, $s2, $t3       # $a0 <- BitVectorAddress
    lw      $a1, 4($s1)         # $a1 <- ToNode
    jal     setBit
    addi    $s1, $s1, 8
    j       nextEdge
Done:
    lw      $ra, 0($sp)
    lw      $s0, 4($sp)
    lw      $s1, 8($sp)
    lw      $s2, 12($sp)
    lw      $s3, 16($sp)
    addi    $sp, $sp, 20
    jr      $ra
```